Phys 238: The Normal Distribution

In[252]:=

```
Clear["Global`*"]
```

In[253]:=

$$f[x\_, X\_, \sigma\_] := \left(\frac{1}{\sigma\, Sqrt[2\,\pi]}\right) Exp\left[-\frac{(x-X)^2}{2\,\sigma^2}\right]$$

In[254]:=

```
f[x, X, σ]
```

Out[254]=

$$\frac{e^{-\frac{(x-X)^2}{2\sigma^2}}}{\sqrt{2\pi}\,\sigma}$$

Here are is a comparison of two distribution functions, with σ=1 and σ=2.

In[255]:=

```
Plot[{f[x, 0, 1], f[x, 0, 2]}, {x, -10, 10}, PlotRange → All, LabelStyle → Larger,
 AxesLabel → {"x", "f(x)"}, PlotLegends → {"σ=1", "σ=2"}, ImageSize → Scaled[0.8]]
```

Out[255]=



Probability functions

For simplicity, take the mean X to be 0. (One could always define a new variable z=x-X.)

In[256]:=

```
X = 0;
```

Normalization

The distribution is already normalized (subject to the reasonable assumption that σ is a real positive number).

In[257]:=
```
Integrate[f[x, X, σ], {x, -∞, ∞}, Assumptions → {σ > 0}]
```
Out[257]=
```
1
```

Function showing the probability of being within ‘t’ standard deviations of the mean.

In[258]:=
```
p[t_] := Integrate[f[x, X, σ], {x, X - t σ, X + t σ}, Assumptions → {σ > 0}]
```
In[259]:=
```
p[1]
```
Out[259]=

$$\mathrm{Erf}\left[\frac{1}{\sqrt{2}}\right]$$

While the Error Function Erf[ ] is technically a correct answer, it isn’t particularly practical. Tell Mathematica to give the numerical approximation:

In[260]:=
```
N[p[1]]
```
Out[260]=
```
0.682689
```

Tabulate the probability of getting a result within 1, 2, 3, 5, or 10 standard deviations.

In[261]:=
```
Table[{t, N[p[t], 10]}, {t, {1, 2, 3, 5, 10}}] // TableForm
```
Out[261]//TableForm=

1	0.6826894921
2	0.9544997361
3	0.9973002039
5	0.9999994267
10	1.000000000

The 95% confidence level corresponds to just under 2σ. (The Abs[] ensures that Mathematica only tries real numbers for t.)

In[262]:=
```
FindRoot[N[p[Abs[t]]] == 0.95, {t, 1.}]
```
Out[262]=
```
{t → 1.95996}
```

The 99% confidence level corresponds to about 2.6 σ.

In[263]:=
```
FindRoot[N[p[Abs[t]]] == 0.99, {t, 1}]
```
Out[263]=
```
{t → 2.57583}
```